
1. Программное обеспечение.
1.1. Информационно-поисковая система .

Работа по автоматизированной подготовке беловой рукописи ведется на основе информационно-поисковой системы (ИПС) "Русская словесность" в среде пакета прикладных программ CDS/ISIS for Wimdows 1.5_3, дополненного оригинальными программными разработками Отдела библиографии и источниковедения ИРЛИ и Отдела информатики БАН (программист В.В. Рубцов).
Создание беловой рукописи в среде CDS/ISIS позволяет автоматически формировать*:
- автоматическую сортировку по заданным параметрам: главам, хронологии и алфавитному порядку (авторов, заглавий)
- тотальную нумерацию записей;

- заголовки глав и разделов глав;

- приложения в виде указателей: именного, предметного, авторов и 

  произведений, заглавий журналов и пр. с проставлением номеров 

  записей. 

*См. "Руководство к ISIS".

1.2. Создание и редактирование записей рекомендуется во внешнем текстовом редакторе (Multi-Edit). Некоторые версии редактора Multi-Edit имеют особенности. 

1.2.1. Буква "я" (строчная):

- не может быть первой буквой в строке,
- будучи последней, уничтожается при сохранении файла,
- при копировании "я" преобразуется в пробел.
1.2.1.1. Временное решение проблемы:
- не делаем слово с начальным или конечным "я" первым или последним в строке,
- вводим «+я» или «я+».
- перед загрузкой в БД окончательной версии данных убираем "+" в Блокноте.

Пример:\608  ^aрецензия+

1.2.1.2. Устанавливаем подпрограммы Ротько В.А. (см. «Поддержка национального языка (NLS) в редакторе Multi-Edit for Windows»).

1.2.2. При сохранении  в первых 117 строках возможно уничтожение пробелов и «я» в конце слов в середине строки. Восстанавливаем в Блокноте.
1.3. Нестандартную графику отображаем при помощи UNICODE. Для удобства поиска в ИПС поисковые ключи помещаем в подполя 9 соответствующих полей без кодов (только опорную букву латинского алфавита). 
Пример: \210  ^aK&ouml;ln^d1994^9Koln
2. Требования к ИПС: наличие технологических средств и их документации, способов и их описания.
2.1. CDS /ISIS for Wimdows1.5_3, см. руководство.

2.2. Набор дополнительных подпрограмм:

· проверки, загрузки, выгрузки данных из текстового файла в формате DAT:

· cor00-.exe (встроены в ИПС): cor001.exe (выгрузка), cor002.exe (загрузка), cor003.exe (проверка полей),
· Psubf.exe (проверка подполей, запускается извне);
· описание технологии их использования: cor000.txt.
2.3. Наличие описания формата данных: Uni-irli.txt или -.doc.

2.4. Описание кодовой системы-рубрикатора указателя.
2.5. Наличие рабочих листов подготовки данных в БД: [имя БД].fmt.

2.6. Наличие форматов вывода записей (см. в Приложении):
· вывода на экран – [имя БД].pft, 

· во внешний файл (для печати) – p.pft ,
· в гипертекстовый формат – htm.pft и htm-6.pft.
2.7. Наличие таблицы выбора полей для создания инвертированного файла базы – словаря (см. в Приложении): [имя БД].fst
2.8. Наличие технологии автоматического формирования и вывода на печать текста указателя (см. "Руководство к ISIS").

3. Требование к базе библиографических данных (БД).*
3.1. Соответствие формату ЮНИМАРК-ИРЛИ (без встроенных полей).
3.2. Соответствие формату внешнего файла DAT.
Первый знак каждой строки файла может содержать только перечисленные знаки: *{%- пустая строка, где
· * начало файла

· { комментарий

· % начало записи

· - продолжение записи

· пустая строка - конец записи.
3.3. Наличие системы кодов записей  для сортировки по разделам указателя.
Каждому разделу соответствует номер. Коды подразделов внутри разделов имеют двучленный номер, где первая цифра связывает подраздел с разделом. 

Пример:

1. Сочинения.
1.1. ПСС.
1.2. СС.
1.3. Избранное.
1.4. Отдельные издания.
1.5. Письма.
2. Литература о нем.
2.1. Биография.
2.2. Литература о творчестве.
3. Справочники.
* На примере БД «Гончаров» (GIA)

4. Проверка данных на соответствие формату DAT перед загрузкой.

4.1. Проверка в Multi-Edit. 
Поиск наличия (копируем поисковое выражение в строку поиска):

· неформатных строк: %[~%~\~@-~{]

· повторных пустых строк: %$%$

· пробелов после знака подполя “^? “

· кириллических букв в идентификаторе подполя: ^[аеср]
· частые ошибки форматирования:

· \2001 ^a?*$\463^
· \2001 ^t
· \225  ^t
· незаполненные подполя: ^?^
4.2. Проверка  корректности данных – формата DAT встроенными средствами ИПС (cor003.exe):

· опция “Работа с файлами DAT” => запустить программу Проверка => выбрать файл,

Рис. 1. Программа проверки
· в каталоге, где лежит проверенный файл, см. файл *.err, 
· открыть проверяемый файл и файл ERR в текстовом редакторе (Multi-Edit или Блокнот),

· исправляем ошибки: копируем номер строки => вставляем в окошко номера строки (в Multi-Edit см. нижнее меню проверяемого файла) => сохраняем результат.
4.3. Проверка через внешние программные средства ИПС на корректность представления подполей в файле:

· запустить программу Psubf в коренном каталоге ISIS,

· выбрать файл,

· выбрать «допускаются буквенные и цифровые подполя»,

· выполнить проверку,

· исправить найденные ошибки во внешнем файле DAT.

Рис. 2. Программа проверки подполей.
5. Загрузка данных в систему.

5.1. Выбрать БД в ИПС. Если БД содержит ненужные данные, ее следует обнулить (опция Утилиты => реинициализировать)

5.2. Опция “Работа с файлами DAT” => запустить программу Загрузка (cor002.exe) => 
· в 1-м поле загрузки найти файл проверенных данных, 

· во 2-м поле найти файл [имя БД].MST выбранной БД (напр., GIA),
· выполнить загрузку. 

Рис. 3. Загрузка файла в систему.

5.2. Активировать базу перелистыванием записей.

6. Проверка данных средствами Поиска ИПС. 

6.1. Поиск ошибок в записях:

- отсутствие поля кодов указателя: ?a(v689^c)

- повторение поля кода указателя: ?nocc(v689)>1 

- код другой системы разделов: ?not v689^a:'код библиографической базы данных', напр.: ?not v689^a:'1gju'

- отсутствие обязательного поля Заглавие: ?a(v200^a)

- отсутствие заглавия в поле Хост: ?p(v463) and a(v463^t)

- отсутствие автора при наличии соавтора: ?p(v701) and a(v700)

- соавтор в поле автор: ?nocc(v700)>1 
- отсутствие основного подполя а поле Автор: ?p(v700) and a(v700^a)

- ошибки в идентификаторе подполя:
- ?p(v225) and a(v225^a)
- ?p(v225^g)
- отсутствие поля физическая характеристика (страницы): ?a(v215) and a(v462 ^p)

6.2. Исправить ошибки во внешнем файле. 
6.3. Повторить загрузку.

7. Сортировка данных.

ИПС обеспечивает сортировку по нескольким параметрам:

- коду главы,

- хронологии (по времени выхода в свет),

- алфавиту (авторов, затем заглавий).

7.1. Обновляем словарь БД.
7.2. Сортируем: опция Печать => Повтор (вызов модуля сортировки) => выбираем HTM.wpr => Сортировка => Печать (в условный файл, напр., 1.txt). См. рис. 4.

Рис. 4. Окно Сортировки записей перед печатью: по разделам, датам издания, автору и/или заглавию с заданными заголовками раздела в виде кодов с htm-тегами.
7.3. Выгружаем отсортированные данные в iso-формат: опция База данных => Экспорт => в условный файл, напр., 1.iso => Из отсортированного HIT-файла.

7.4. Загружаем (База данных => Импорт) 1.iso обратно в режиме перезаписи.

Подобная сортировка позволяет использовать номер записи в БД в качестве номера записи во внешней форме – беловой рукописи, гипертекстовой системе и т.д. и обеспечивает систему внутренних ссылок.

Формат заголовка обеспечивает связь с файлом, содержащим гиперссылки на заголовки разделов: 

##,'<h2><a="gia-cod.htm#'v1'" id="'v1,'">'v1,'</a></h2>'#

8.Формирование словаря - инвертированного файла БД 
Поля и подполя данных, отражаемых в словаре определяются таблицей выбора полей - файл [имя БД].FST (напр., gia.FST – см. приложение).

Меню ИПС: База данных => Обновление инвертированного файла => создать новый словарь.

9. Создание ссылок на записи внутри базы. Для создания в выгрузке гиперссылки внутри текста: «См. № …» и  «См. также № …».
9.1. Ссылка на другую запись в этой же базе данных обеспечивается ссылкой из поля 98^a записи 1 к полю 90^b записи 2 через представление данных о последней в словаре: (|090:|v090^b/)
9.2. Ссылка из записи 1 на запись 2 в той же базе данных обеспечивается через отражение номера (mfn) записи 2 в поле 98^b.
9.3. Строка в формате выгрузки:
('<a href="#r' v098^a,|: |v098^b,|, |v098^c, '">' |, |+v098^a,|: |v098^b,|, |v098^c '</a>')

дает нам
<p class="other1">См. также № <a href="#r23">23</a> .</p>

или если повтор номера (поля 098а)

<p class="other1">См. также № <a href="#r47">47</a>, <a href="#r52">52</a>.</p>

10. Выгрузка на печать.
10.1. Сортируем базу (для отражения кодов разделов) => Повтор => модуль сортировки HTM.wpr.
10.2. «Общая» страница печати:
- проверить количество записей  интервал MFN или вся БД,
=> вывод во внешний файл,

=> печать,
=> выбрать формат вывода во внешний файл или HTM* (внизу слева – «список файлов типов»),
=> задать название файла (gia),

=> отправить на печать – «Да».
*см. приложение

Рис. 5. Окно сохранения файла выгрузки.

11. Создание приложений – вспомогательных указателей.
Для выгрузки приложений из инвертированного файла необходимы таблицы выбора полей *.fst (name.fst и pr-all.fst – см. приложение)
11.1. В окне Печать – Повтор выбираем соответствующий модуль сортировки: pril.wpr

[image: image6]
Рис. 6. Окно выбора модуля сортировки.

11.2. В опции Сортировка:

- задаем формат заголовка ##,'<p class="record2">'v1,'<br>'#
- Ключ 1: вписываем название формата выгрузки


- @name - персонально-авторский указатель,


- @pr-all – тематические рубрики, произведения без автора, периодические издания.

11.3. Общая страница печати: формат печати:
('<a href="gia.htm#r',f(mfn,0,0),'">',f(mfn,0,0),'</a>,'),/
11.4. Печать в выходной файл => в формат HTM* (внизу слева – «список файлов типов»).

12. Создание гипертекстовой системы 

12.1. В систему входит:

- стилевой лист,
- стартовая страница, содержащая ссылки на все файлы системы,

- массив библиографических документов,
- список разделов указателя,
- именной указатель,
- предметный указатель.
12.2. Данные об оформлении документа заданы стилевым листом - sample.css, ссылка на который должна содержаться в информации о файлах (в заголовке):
<link rel="STYLESHEET" type="text/css" href="sample.css">
12.3. Создание списка разделов указателя в гипертекстовом формате:

12.3.1. Выгрузка списка разделов указателя с пустыми записями из базы.
12.3.2. Замена в ручном режиме цифровых кодов на названия разделов.
12.4.Работа с гипертекстовым массивом библиографических документов.

12.4.1. В каталоге WORK найти файлы выгрузки с гипертекстовой разметкой (gia.htm, name.htm, pr.htm).
12.4.2. Открыть их в любой программе, позволяющей обработать гипертекстовый файл (Блокнот, МультиЭдит, Notepad++).
12.4.3. Скопировать из файлов 0-*.htm (0-gia.htm, 0-name.htm, 0-pr.htm) заголовки гипертекстовых файлов с данными о стиль-листе sample.css, ссылками на другие файлы системы и окончание файлов. 

12.4.4. Коды разделов и подразделов поменять на их названия с соответствующими тегами. Пример:
- в выгрузке
<h2><a ="gia-cod.htm#1" id="1">1</a></h2>

- в результате
<h2><a ="gia-cod.htm#1" id="1">Раздел I. Публикации произведений И.А. Гончарова</a></h2><br>
12.4.5. Если в данных имеются коды нестандартной графики, поменять их на UNICODE. 
12.4.7. Удалить в режиме поиск-замена:

- лишние точки (одну из удвоенных, восстанавив третью точку 

в многоточии),

- пробелы перед знаками препинания.
12.5. Ссылки между файлами системы.

В начало каждого выгруженного или созданного текста включаем ссылку на остальные файлы гипертекстовой системы, что обеспечивает переход в нужный файл:
 <a href="gia-cod.htm" target="_top">Разделы библиографического указателя</a></td>

<a href="art.htm" target="_top">Введение</a></td>

<a href="gia.htm" target="_top">Текст библиографического указателя</a></td>

<a href="name.htm" target="_top">Именной указатель</a></td>

<a href="pr.htm" target="_top">Предметный указатель</a></td>

<br>
13. Подготовка электронного текста беловой рукописи для представления в издательство. 
Для верстки необходимо представить текст:

1) в виде бумажной копии беловой рукописи указателя с форматированием (колонтитулы, абзацы, выделения шрифтом),

2) в виде электронного текста без форматирования (абзацы отделяются пустой строкой).

Формат выгрузки электронного файла должен быть согласован с издательством. Для автоматизации процесса верстки после консультации с верстальщиком издательства текст размечается специальными знаками (что упрощает процесс верстки и уменьшает ошибки).

Приложения

Таблица выбора полей для словаря - GIA.FST
90 0 (|090:|v090^b/)

92 0 (|bibl:|v92^a,| |v92^b/)

200 0 (v200^a,|: |v200^e/)

200 0 (|200:|v200^a,|: |v200^e/)

200 0 (v200^9/)

210 0 (|210a:|v210^a/)

210 0 (|210c:|v210^c/)

210 0 (|210c:|v210^c/)

210 0 (|2109:|v210^9/)

321 0 (v321^a/)

327 0 (v327^a/)

461 0 (v461^a,| |v461^t/)

461 0 (v461^t/)

461 0 (v461^9/)

463 0 (v463^a,| |v463^t/)

463 0 (v463^t,|: |v463^e/)

463 0 (v463^9/)

470 0 (v470^a,| |v470^t/)

470 0 (v470^t,|: |v470^e/)

470 0 (v470^t,|: |v470^e/)

470 0 (v470^9/)

488 0 (v488^t,|: |v488^e/)

488 0 (v488^9/)

600 0 (v600^a,| |v600^b,|, |v600^c,| |v600^d,|, |v600^g,| - |v600^x/)

600 0 (v600^9/)

600 0 (|600:|v600^a,| |v600^b,|, |v600^g,| |v600^d,|, |v600^c/)

601 0 (v601^a,| |v601^b,| |v601^c,| |v601^9/)

601 0 (v601^9/)

601 0 (|601:|v601^a,| |v601^b,| |v601^c/)

602 0 (|602:|v602^a/)

604 0 (v604^a,| “|v604^t|”|/)

604 0 (v604^9/)

604 0 (|604:|v604^a,| "|v604^t|"|/)

605 0 (v605^a,| - |v605^b/)

605 0 (v605^9/)

605 0 (|605:|v605^a,| - |v605^b/)

606 0 (v606^a/)

606 0 (v606^z/)

606 0 (|606:|v606^a/)

607 0 (|607:|v607^a,| |v607^b,|, |v607^y/)

608 0 (v608^a/)

608 0 (|608:|v608^a/)

610 0 (v610^a/)

610 0 (|610:|v610^a/)

661 0 (v661^a/)

689 0 (|cod:|v689^c/)

700 0 (v700^a,| |v700^b,| |v700^c,| |v700^g,| |v700^d/)

700 0 (|700:|v700^a,| |v700^b,| |v700^c,| |v700^g,| |v700^d/)

700 0 (v700^9/)

701 0 (v701^a,| |v701^b,| |v701^c,| |v701^g,| |v701^d/)

701 0 (|701:|v701^a,| |v701^b,| |v701^c,| |v701^d,| |v701^g/)

701 0 (v701^9/)

702 0 (v702^a,| |v702^b,|, |v702^c,| |v702^d,| |v702^g/)

702 0 (|702:|v702^a,| |v702^b,|, |v702^c,| |v702^d,| |v702^g/)

702 0 (v702^9/)
Формат вывода на экран

gia.PFT

title('Г О Н Ч А Р О В  И.  А.  1959-2005'), bpict(('gonch.jpg'),('gonch2.jpg'),0), f2,cl1, {m(800),link((fs20,b,'О базе данных'),'TEXTBOXRCHILDLOAD' 'FORMAT 1gia'), tab,link((fs20,b,'О поиске'),'TEXTBOXRCHILDLOAD' 'FORMAT 2gia'), if p(v099) then tab, {cl2,i,b,'См. полный текст:  '} (link((fs20,b,v099^b+|;|), 'OPENFILE ', v099^a)'   ')fi, /{fs20,b,cl2,'Форматы вывода данных:  '}, link((fs20,b,'для Word;'),'TEXTBOXRCHILDLOAD' 'FORMAT Word'), {fs20,b,cl2,'   для принтера '},link((fs20,b,'"GIAprn",'),'TEXTBOXRCHILDLOAD' 'FORMAT giaprn')'   ',link((fs20,b,'"P"'),'TEXTBOXRCHILDLOAD' 'FORMAT p')/, {cl12,i,'Раздел: '}{@kod}}/ fs26, @fld,

fld.PFT
{tab,{b,if v700.2:'1' then if p(v700^a) and p(v700^b) then v700^a||,| |v700^b| |, fi, if p(v700^a) and a(v700^b) then if p(v700^g) then v700^a,| |v700^g|. |, else v700^a|. | fi, fi, if p(v700^d) then ' ',v700^d,fi, if p(v700^c) then ', ',v700^c|. |,fi,fi, if v700.2:'0' then (v700^a||,| |v700^b,|, |v700^c,| |v700^d)'. ', fi, if (p(v700) and a(v701)) then 

else (|, |v701^a,| |v701^b| |,v701^d,|, |v701^c),fi'',

if (p(v463) or p(v461)) then if (a(v700) and p(v200)) then  fi, 

if p(v461) then v461^t,(": "v461^e+|. |),  if (a(v461^f) and p(v461^g)) then ' / '(v461^g+|; |) else | / |v461^f,(|; |v461^g) fi, |. - |v205^a, |. - |v461^c,|, |v461^d,|. - |v461^v'.',

if p(v225) then | - (|v225^a,(| = |v225^b),(|: |v225^e), if p(v225^f) then ' / ' (v225^f+|; |) fi,(|, |v225^x),|; |v225^h,|. |v225^i,|; |v225^v, if p(v225) then ').'fi fi,

/tab, (|. |+v200^a||),(| = |v200^d),(|: |v200^e), if (a(v200^f) and p(v200^g)) then (| / |+|; |v200^g) else | / |v200^f,(|; |+v200^g), fi, if (p(v200^h) and p(v200^i)) then | - |v200^h,|: |v200^i,'.' fi, if (a(v200^h) and p(v200^i)) then | - |v200^i,'.' fi, 

if p(v210) then '. - '(v210^a+|; |),|, |d210^a,v210^d,'.' fi, 

if (p(v461^h) or p(v461^i)) then | - |v461^h,|: |v461^i,'.' fi,| - |v215^a|.| fi, if p(v463) then (|. |+v200^a||),(| = |v200^d),(|: |v200^e), if (a(v200^f) and p(v200^g)) then (| / |v200^g+|; |) else | / |v200^f, (|; |+v200^g), fi, ' // ' (|, |+v463^a), if p(v463^a) then ' ' fi,v463^t, (| = |v463^d),(|: |v463^e), if p(v463^c) then '; '(|; |+v463^c) fi, if (a(v463^f) and p(v463^g)) then ' / '(v463^g+|; |) else | / |v463^f, "; "v463^g, fi, '.', if (p(v463^i) and p(v463^h)) then | - |v463^h,|. |v463^i,'.' fi, if (a(v463^i) and p(v463^h)) then | - |v463^h,'.' fi, if (p(v463^i) and a(v463^h)) then | - |v463^i,'.' fi, 

if p(v205) then | - |v205^a,(| = |v205^d), if p(v205^f) then ' / '(v205^f+|; |) fi, if p(v205^b) then (|, |v205^b) fi, '.' fi, 

if v210<>'' then ' - '(|; |+v210^a,if v210^c<>'' then |: |v210^c fi), if (p(v210^a) or p(v210^c)) then ', ' fi, v210^d,'.'fi,

if p(v463^v) then | - |v463^v|.| fi,| - |v215^a|.| fi, else 

if p(v710^a) then v710^a,|:|v710^e,| (|v710^k||,|; |v710^j|)|,|. |v710^c|.| fi, 

if (a(v700) and p(v200)) then fi,(|. |+v200^a||), (| = |v200^d),(|: |v200^e), if p(v200^c) then '; '(|; |+v200^c) fi, if (a(v200^f) and p(v200^g)) then ' / ' (|; |+v200^g) else | / |v200^f,(|; |v200^g) fi '.', if (p(v200^h) and p(v200^i)) then | - |v200^h,|: |v200^i,'.'fi, if (a(v200^h) and p(v200^i)) then | - |v200^i,'.' fi, 

if p(v205) then | - |v205^a,(| = |v205^d),if p(v205^f) then ' / '(v205^f+|; |) fi, if p(v205^b) then (|, |v205^b) fi, '.' fi,

if v210<>'' then ' - '(|; |+v210^a,if v210^c<>'' then |: |v210^c fi), if (p(v210^a) or p(v210^c)) then ', ' fi, v210^d,'.'fi, if (p(v210^e) or p(v210^g) or p(v210^h)) then ' (',(v210^e+|; |), if p(v210^e) and p(v210^g) then '; ',(v210^g+|; |) fi, if a(v210^e) and p(v210^g) then (v210^g+|; |) fi,(v210^h|, |),')' '.' fi,

if p(v200^v) then | - |v200^v,'.' fi,

if p(v215) then ' - ',v215^a, if p(v215^a) and p(v215^c) then ': ' v215^c fi , if p(v215^d) then '; ' v215^d||,| + |v215^e, fi , fi, '.' fi, 

if p(v225) and a(v461) then (| - (|v225^a,| = |v225^b,|: |v225^e,if p(v225^f) then ' / ' v225^f+|; | fi, |, |v225^x,|; |v225^h,|. |v225^i,|; |v225^v, if p(v225) then ').' fi), fi,

if p(v300) then ' - Примеч.: '(| - |+v300^a|.|) fi,

if p(v320) then (| - |v320^a|.|) fi},

(if v462<>'' then  /tab if v462^a<>'' and v462^t<>'' then v462^a|. |,v462^t, else v462^a,v462^t, fi, | / |v462^g,|. - |v462^c|,|,|. - |v462^d,|. - |v462^p|.|, fi),

if p(v324) then / tab, (v324^a|.|) fi,

if p(v541^a) then / tab, |Перевод загл.: |v541^a fi,

if p(v327) then /tab, if v327.0:'0' then 'Из содерж.: ' (v327^a+|; |), else 'Содерж.: '(v327^a+|; |)fi,'.' fi,

(if v330^a<>'' then /tab,'Аннотация: 'v330^a|.| fi),

if p(v321) then /tab, {cl4("Рец.: "v321^a+|; |)}'.' fi,

if v098<>'' then /tab,'См. № ', (link((|;  |+v098^a,), 'TEXTBOXRCHILDLOAD 098,xxxx,'|090:|v098^a)), fi'.',

{cl3, if p(v470) then if v470.2:'1' then /tab 'Рец. на кн.: ' fi,(|; |+v470^a if p(v470^a) then ' ' fi, v470^t||,if p(v470^g) then ' / 'v470^g fi, if p(v470^n) then ' // 'v470^n fi,|. - |v470^h|.|, | |v470^i|.|, |. - |v470^v, |. - |v470^e, if p(v470^c) then |. - |v470^c,|, |v470^d, else |. - |v470^d,fi,|. - |v470^p+|; |) ,'.' fi,

if v488.2:'1' then / tab 'См. также: ', (v488^0|: |, v488^a| |,v488^t, if p(v488^g) then ' / 'v488^g fi, if p(v488^n) then ' // 'v488^n fi,|. - |v488^h|.|, | |v488^i|.|, |. - |v488^v, |. - |v488^e, if p(v488^c) then |. - |v488^c,|, |v488^d, else |. - |v488^d,fi,|. - |v488^p+|; |),'.' fi,

if p(v454) then /tab 'Оригинал: ', (v454^0|: |, v454^a| |,v454^t, if p(v454^g) then ' / 'v454^g fi, if p(v454^n) then ' // 'v454^n fi,|. - |v454^h|.|, | |v454^i|.|, |. - |v454^v, |. - |v454^e, if p(v454^c) then |. - |v454^c,|, |v454^d, else |. - |v454^d,fi,|. - |v454^p+|; |),'.' fi}}

(if v333^a<>'' then /tab,'Примеч. для ред.: 'v333^a|.| fi),

{if p(v600) or p(v601) or p(v602) or p(v604) or p(v605) or p(v606) or p(v607) or p(v608) or p(v610) then /cl12,b,f2,i,@6 fi}, if p(v702) then /tab 'Редакторы, составители и пр.: '(|; |+v702^a,| |v702^b| |,v702^d,|, |v702^c,| - |v702^0),fi, /#{fs20,b,cl11,'Источник: '(|; |+v090^a,|. Запись |v090^b)}, if p(v091) then /{fs20,cl0,'Б-ка, шифр: '(|; |+v091^a,|, |v091^b)} fi,/

Формат вывода на печать
P.PFT (с блоком поисковых полей)

f2,

if p(v689) then {fs20,i,'Код в указателе: '(v689^a|:|,v689^c+| / |)fi/}, 

fs26,

{tab,{b,if v700.2:'1' then if p(v700^a) and 

p(v700^b) then v700^a||,| |v700^b| |, fi,

if p(v700^a) and a(v700^b) then if p(v700^g) then v700^a,| |v700^g|. |, else v700^a|. | fi, fi,

if p(v700^d) then ' ',v700^d,fi, if p(v700^c) then ', ',v700^c|. |,fi,fi, 

if v700.2:'0' then (v700^a||,| |v700^b,|, |v700^c,| |v700^d)'. ', fi,

if (p(v700) and a(v701)) then 

else (|, |v701^a,| |v701^b| |,v701^d,|, |v701^c),fi'',

if (p(v463) or p(v461)) then if (a(v700) and p(v200)) then  fi, 

if p(v461) then v461^t,(": "v461^e+|. |),  if (a(v461^f) and p(v461^g)) then ' / '(v461^g+|; |) 

else | / |v461^f,(|; |v461^g) fi, |. - |v205^a, |. - |v461^c,|, |v461^d,|. - |v461^v'.',

if p(v225) then | - (|v225^a,(| = |v225^b),(|: |v225^e),

if p(v225^f) then ' / ' (v225^f+|; |) fi,(|, |v225^x),|; |v225^h,|. |v225^i,|; |v225^v, 

if p(v225) then ').'fi fi,

/tab, (|. |+v200^a||),(| = |v200^d),(|: |v200^e), 

if (a(v200^f) and p(v200^g)) then (| / |v200^g+|; |) else | / |v200^f,(|; |+v200^g), fi, 

if (p(v200^h) and p(v200^i)) then | - |v200^h,|: |v200^i,'.' fi, 

if (a(v200^h) and p(v200^i)) then | - |v200^i,'.' fi, 

if p(v210) then '. - '(v210^a+|; |),|, |d210^a,v210^d,'.' fi, 

if (p(v461^h) or p(v461^i)) then | - |v461^h,|: |v461^i,'.' fi,| - |v215^a|.| fi, 

if p(v463) then (|. |+v200^a||),(| = |v200^d),(|: |v200^e), 

if (a(v200^f) and p(v200^g)) then (| / |v200^g+|; |) else | / |v200^f, (|; |+v200^g), fi, ' // ' (|, |+v463^a), 

if p(v463^a) then ' ' fi,(|. |+v463^t), (| = |v463^d),(|: |v463^e), 

if p(v463^c) then '; '(|; |+v463^c) fi,

if (a(v463^f) and p(v463^g)) then ' / '(v463^g+|; |) else | / |v463^f, "; "v463^g, fi, '.',

if (p(v463^i) and p(v463^h)) then | - |v463^h,|. |v463^i,'.' fi, 

if (a(v463^i) and p(v463^h)) then | - |v463^h,'.' fi, 

if (p(v463^i) and a(v463^h)) then | - |v463^i,'.' fi, 

if p(v205) then | - |v205^a,(| = |v205^d), if p(v205^f) then ' / '(

v205^f+|; |) fi, if p(v205^b) then (|, |v205^b) fi, '.' fi, 

if v210<>'' then ' - '(|; |+v210^a,if v210^c<>'' then |: |v210^c fi),

 if (p(v210^a) or p(v210^c)) then ', ' fi, v210^d,'.'fi,

if p(v463^v) then | - |v463^v|.| fi,| - |v215^a|.| fi, else 

if p(v710^a) then v710^a,|:|v710^e,| (|v710^k||,|; |v710^j|)|,|. |v710^c|.| fi, 

if (a(v700) and p(v200)) then fi,(|. |+v200^a||), (| = |v200^d),(|: |v200^e),

if p(v200^c) then '; '(|; |+v200^c) fi, if (a(v200^f) and p(v200^g)) then ' / '

(|; |+v200^g) else | / |v200^f,(|; |v200^g) fi '.', 

if (p(v200^h) and p(v200^i)) then | - |v200^h,|: |v200^i,'.'fi, 

if (a(v200^h) and p(v200^i)) then | - |v200^i,'.' fi,

if p(v205) then | - |v205^a,(| = |v205^d),if p(v205^f) then ' / '(v205^f+|; |) fi, 

if p(v205^b) then (|, |v205^b) fi, '.' fi,

if v210<>'' then ' - '(|; |+v210^a,if v210^c<>'' then |: |v210^c fi),

 if (p(v210^a) or p(v210^c)) then ', ' fi, v210^d,'.'fi,

if (p(v210^e) or p(v210^g) or p(v210^h)) then ' (',(v210^e+|; |), 

if p(v210^e) and p(v210^g) then '; ',(v210^g+|; |) fi, 

if a(v210^e) and p(v210^g) then (v210^g+|; |) fi,(v210^h|, |),')' '.' fi,

if p(v200^v) then | - |v200^v,'.' fi,

if v215^a<>'' then ' - ',v215^a, if p(v215^a) and p(v215^c) then ': ' v215^c ,

'.' fi , fi, fi,

if p(v225) and a(v461) then (| - (|v225^a,| = |v225^b,|: |v225^e, if p(v225^f) 

then ' / ' v225^f+|; | fi, |, |v225^x,|; |v225^h,|. |v225^i,|; |v225^v, if p(v225) then ').' fi), fi,

if p(v300) then ' - Примеч.: '(| - |+v300^a|.|) fi,

if p(v320) then (| - |v320^a|.|) fi,

if p(v541^a) then | - Перевод загл.: |v541^a fi},

(if v462<>'' then  /tab if v462^a<>'' and v462^t<>'' then v462^a|. |,v462^t, else v462^a,v462^t, fi,

| / |v462^g,|. - |v462^c|,|,|. - |v462^d,|. - |v462^p,'.' fi),

if p(v324) then /tab, (v324^a|.|) fi,

if p(v327) then /tab, if v327.0:'0' then 'Из содерж.: ' (v327^a+|; |)'.', 

else 'Содерж.: '(v327^a+|; |)'.' fi, fi,

(if v330^a<>'' then /tab,'Аннотация: 'v330^a|.| fi),

if p(v321) then /tab, {("Рец.: "v321^a+|; |)} fi,

{if p(v470) then /tab 'Рец. на кн.: ' fi,(|; |+v470^a 

if p(v470^a) then ' ' fi, v470^t||,if p(v470^g) then ' / 'v470^g fi,

if p(v470^n) then ' // 'v470^n fi,|. - |v470^h|.|, | |v470^i|.|, |. - |v470^v, |. - |v470^e,

if p(v470^c) then |. - |v470^c,|, |v470^d,

else |. - |v470^d,fi,|. - |v470^p|.|, | - диакритика: |v470^9+|; |),

if p(v488) then /tab 'См. также: ', (|; |+v488^a, 

if p(v488^a) then ' ' fi, v488^t||, if p(v488^g) then ' / 'v488^g fi,

if p(v488^n) then ' // 'v488^n fi,|. - |v488^h|.|, | |v488^i|.|, |. - |v488^v, |. - |v488^e, 

if p(v488^c) then |. - |v488^c,|, |v488^d,

else |. - |v488^d,fi,|. - |v488^p|.|, | - диакритика: |v488^9+|; |),fi}}

{if p(v600) or p(v601) or p(v602) or p(v604) or p(v605) or p(v606) or p(v607) or p(v608) 

or p(v610) then cl12,fs22,@6 fi}

if p(v091) then /tab, {fs20,cl0,'Б-ка, шифр: '(|; |+v091^a,|, |v091^b)} fi,

if p(v333) then /tab, {fs20,i,'Примеч. для библиогр., ред.: '}{fs20,b,(v333^a/)} fi,

/#
Формат вывода в гипертекстовый файл

HTM.PFT(с блоком поисковых полей)
'<div class="r" id="r'mhl,f(mfn,1,0)'">',

'<p class="title">',

'<span class="number">'mhl,f(mfn,1,0),'. ','</span>',

'<span class="author"><i>'

if v700.2:'1' then if p(v700^a) and p(v700^b) then

 v700^a||,| |v700^b||, fi,

if p(v700^a) and a(v700^b) then v700^a|. |, fi,

if p(v700^d) then ' ',v700^d,fi,

if p(v700^c) then ', ',v700^c|. |,fi,fi,| |v700^0|.|

if v700.2:'0' then

 (v700^a||,| |v700^b,|, |v700^c,| |v700^d,| |v700^0)'.', fi,' ','</i></span>',

'<span class="opus">'

if (p(v463) or p(v461)) then if (a(v700) and p(v200)) then  fi,

 if p(v461) then v461^t,(": "v461^e+|. |),

if (a(v461^f) and p(v461^g)) then ' / '(v461^g+|; |)

 else | / |v461^f,(|; |v461^g+|; |) fi, |. – |v461^c,|, |v461^d,'.',

 if p(v225) then |– (|v225^a,(| = |v225^b),(|: |v225^e),

 if p(v225^f) then ' / ' (v225^f+|; |) fi,(|, |v225^x),|; |v225^v,|. |v225^i,|; |v225^h,

 if p(v225) then ').'fi, fi,| – |v461^v|.|/

(|. |+v200^a),(| = |v200^d),

(|<span class="subhead">: |v200^e|</span>|),

|</span>|,

if (a(v200^f) and p(v200^g)) then (|<span class="a"> / |v200^g+|; |)

 else |<span class="a"> / |v200^f,(|; |+v200^g), '</span>' fi,

if (p(v200^h) and p(v200^i)) then |<span class="a"> – |v200^h,|: |v200^i,'</span>.' fi,

if (a(v200^h) and p(v200^i)) then |<span class="a"> – |v200^i,'</span>.' fi,'</a>',

|.<span class="1"> – |v205^a|</span>|,

if p(v210) then '<span class="place">. – '(v210^a+|; |),|, |d210^a|</span>|,

|<span class="date">|v210^d||,'.</span>' fi,

if (p(v461^h) or p(v461^i)) then | – |v461^h,|: |v461^i,'.' fi,

|<span class="pages"> – |v215^a|.</span>|fi,

if p(v463) then

(|. |+v200^a||),(| = |v200^d), '</span>',

(|<span class="subhead">: |v200^e|</span>|),

if (p(v200^f) or p(v200^g)) then '<span class="a">'

if (a(v200^f) and p(v200^g)) then ' / ' (v200^g+|; |) else | / |v200^f, |; |v200^g, fi,'</span>' ,fi,'</a>',

'<span class="host"> // ' (|, |+v463^a), if p(v463^a) then ' ' fi,v463^t,

 (| = |v463^d),(|: |v463^e), if p(v463^c) then '; '(|; |+v463^c) fi,

if (a(v463^f) and p(v463^g)) then ' / '(v463^g+|; |) else | / |v463^f, "; "v463^g, fi, '.',

if (p(v463^i) and p(v463^h)) then | – <span class="part">|v463^h,|. |v463^i,'.</span>' fi, if (a(v463^i) and p(v463^h)) then | – <span class="part">|v463^h,'.</span>' fi,

if (p(v463^i) and a(v463^h)) then | – <span class="part">|v463^i,'.'  fi,'</span>',

if p(v205) then |<span class="1"> – |v205^a,(| = |v205^d), if p(v205^f) then ' / '(v205^f+|; |) fi, if p(v205^b) then (|, |v205^b) '</span>'fi, '.' fi,

if v210<>'' then '<span class="place"> – '(|; |+v210^a'</span>',

if v210^c<>'' then '<span class="publisher">: 'v210^c fi),

 if (p(v210^a) or p(v210^c)) then ', ' fi, '<span class="date">'v210^d'.</span>'fi,fi,

if p(v463^v) then | – <span class="vol">|v463^v|</span>.| fi,

| – <span class="pages">|v215^a|.</span>| ,

 else if p(v710^a) then |<span class="author">|v710^a,|: |v710^e,| (|v710^k||,|; |v710^j|)|,|. |v710^c|.</span>| fi,

if (a(v700) and p(v200)) then  fi,

(|<span class="opus">||. |+v200^a||), (| = |v200^d),'</span>',(|<span class="subhead">: |v200^e|</span>|,)

 if p(v200^c) then | <span class="a">|(|; |+v200^c|</span>|)  fi,

if (p(v200^f) or p(v200^g)) then '<span class="a">'

if (a(v200^f) and p(v200^g)) then ' / ' (|; |+v200^g) else | / |v200^f, |; |v200^g, fi,'</span>' ,fi,

if (p(v200^h) and p(v200^i)) then | – <span class="part">|v200^h,|: |v200^i,'</span>' fi, if (a(v200^h) and p(v200^i)) then | – <span class="part">|v200^i,'</span>', fi,'.',

if p(v205) then |<span class="1">. – |v205^a,(| = |v205^d),if p(v205^f) then ' / '(v205^f+|; |) fi,if p(v205^b) then (|, |v205^b) fi, '.</span>' fi,

if v210<>'' then '<span class="place"> – '(|; |+v210^a'</span>',

if v210^c<>'' then '<span class="publisher">: 'v210^c fi),

 if (p(v210^a) or p(v210^c)) then ', ' fi, '<span class="date">'v210^d'.</span>'fi,

if (p(v210^e) or p(v210^g) or p(v210^h)) then ' (',(v210^e+|; |),

if p(v210^e) and p(v210^g) then '; ',(v210^g+|; |) fi,

if a(v210^e) and p(v210^g) then (v210^g+|; |) fi,(v210^h|, |),')' '.' fi,

if p(v200^v) then |<span class="vol"> – |v200^v,'.</span>' fi,

if p(v215^a) then '<span class="pages"> – ',v215^a,

if p(v215^a) and p(v215^c) then ': ' v215^c fi '.</span>' fi, fi,

if p(v225) and a(v461) then (|<span class="seria"> – (|v225^a,| = |v225^b,|: |v225^e, if p(v225^f) then ' / ' v225^f+|; | fi, |, |v225^x,|; |v225^h,|. |v225^i,|; |v225^v, if p(v225) then ').</span>' fi), fi,

if p(v300) then '<span class="1"> – Примеч.: '(| – |+v300^a|.|)' </span>', fi,

if p(v320) then '<span class="1"> – '(|; |+v320^a|.|) '</span>', fi,

if p(v541^a) then |<span class="1"> – Перевод загл.: |v541^a '.</span>', fi,'</p>',

(if v099<>'' then /'<p class="other1">См. текст: ', '<a href="' v099^a, '">'   |, |+v099^b, '</a>''.</p>' fi),

 (if v462<>'' then /'<p class="set">' if v462^a<>'' and v462^t<>'' then v462^a| |,v462^t, else v462^t, fi,

 if a(v462^f) and v462^g<>'' then ' / '

v462^g, else | / |v462^f, |; |v462^g, fi 

if v462^c<>'' and v462^d<>'' then |. – |v462^c,|, |v462^d, fi, 

if a(v462^c) and v462^d<>'' then |. – |v462^d, fi, |. – |v462^p,'.', '</p>', fi),

if p(v324) then /'<p class="reprint"><i>Репринт кн.: </i>'(v324^a|.|)'</p>', fi,

if p(v327) then if v327.0:'0' then /'<p class="content"><i>Из содерж.:</i> ' (v327^a+|; |)'.',

 else /'<p class="content"><i>Содерж.:</i> '(v327^a+|; |)'.''</p>',fi fi,

(if v330^a<>'' then /'<p class="abstract">'v330^a|.|'</p>', fi),

if p(v321) then /'<p class="review"><i>Рец.:</i> '(v321^a+|; |)'.</p>', fi,

if v470.2:'1' then/ '<p class="other"><i>Рец. на кн.:</i> ' (|; |+v470^a

if p(v470^a) then ' ' fi, v470^t||,if p(v470^g) then ' / 'v470^g fi,

if p(v470^n) then ' // 'v470^n fi,|. – |v470^h|.|, | |v470^i|.|,  |. – |v470^e,

if p(v470^c) then |. – |v470^c,|, |v470^d,

 else |. – |v470^d,fi,|. – |v470^v,|. – |v470^p),'.</p>',fi,

if v488.2:'1' then / '<p class="other"><i>См. также:</i> ' (|; |+v488^a

if p(v488^a) then ' ' fi, v488^t||,if p(v488^g) then ' / 'v488^g fi,

if p(v488^n) then ' // 'v488^n fi,|. – |v488^h|.|, | |v488^i|.|,  |. – |v488^e,

if p(v488^c) then |. – |v488^c,|, |v488^d,

 else |. – |v488^d,fi,|. – |v488^v,|. – |v488^p),'.</p>',fi,

if v098<>'' then /'<p class="other1">См. также № ', 

('<a href="#r' v098^b, '">'   |, |+v098^b  '</a>')'.</p>' fi,

if p(v600) or p(v601) or p(v602) or p(v604) or p(v605) or p(v606) or p(v607) or p(v608) or p(v610)  or p(v702) then /'<p class="index"><i>Ключевые слова:</i> ',

if p(v600) then (|; |+v600^a,| |v600^b,| |v600^d,|, |v600^c,| – |v600^0|.|) fi ,/

if p(v702) then (|; |+v702^a,| |v702^b,| |v702^d,|, |v702^c,| – |v702^0|.|) fi ,/

if p(v601) then '; ' (|; |+v601^a,| |v601^b,|, |v601^c,|, |v601^x,|, |v601^y,| (|v601^z|)|),  fi,/

if p(v604) then '; ' (|; |+v604^a,| "|v604^t|"|), fi,/

if p(v605) then '; ' (|; |+v605^a,|, |v605^b,| (|v605^f|)|), fi,/

if p(v606) then '; ' (|; |+v606^a,| – |v606^z),  fi,/

if p(v608) then '; ' (|; |+v608^a), fi,/

if p(v607) then '; ' (|; |+v607^a,| |v607^b,|, |v607^y), fi,/

if p(v610) then '; ' (|; |+v610^a) fi,/ '</p>', fi,

(if v090<>'' then /'<p class="source">Источник: 'v090^a,|. № |v090^b'</p>' fi),

'</div>',/#
Форматы выгрузки вспомогательных указателей

Именной: авторы, соавторы, редакторы и пр., персоналии, родовые имена - NAME.FST
1 0 (if v600^a<>'' then v600^a,| |v600^b, if a(v600^b) then | |v600^g fi,| |v600^d,|, |v600^c,|, |v600^x,| |v600^0/ fi)

(v604^a,| "|v604^t|"|,|, |v604^l, if (p(v604^y) or p(v604^k)) then ' (',v604^y, if (p(v604^y) and p(v604^k)) then ', ' fi v604^k')' fi/)
(v700^a,| |v700^b,if a(v700^b) then | |v700^g fi,| |v700^d,|, |v700^c,| |v700^0/)
(v701^a,| |v701^b,if a(v701^b) then | |v701^g fi,| |v701^d,|, |v701^c,| |v701^0/)
(v702^a,| |v702^b,if a(v702^b) then | |v702^g fi,| |v702^d,|, |v702^c,| |v702^0/)
(v602^a/)
Предметный: организации, мероприятия, периодические издания, произведения без авторов, предметные рубрики, жанры описываемых документов,  ключевые слова - PR-ALL.FST
1 0 (v601^a,|, |v601^b,|, |v601^y/)(v601^9/)
(|"|v605^a|"|,|, |v605^b,|, |v605^l, if (p(v605^y) or p(v605^k)) then ' (',v605^y if (p(v605^y) and p(v605^k)) then ', ' fi v605^k')' fi/)(v605^9/)

